/*
 * Copyright 2006-2007 the original author or authors.
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *      http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */
package asphalt2.util
{
	import asphalt2.container.PageContainer;
	
	import flash.display.DisplayObject;
	import flash.events.Event;
	import flash.events.IEventDispatcher;
	import flash.geom.Point;
	
	import mx.controls.Alert;
	import mx.core.Application;
	import mx.core.Container;
	import mx.core.IFlexDisplayObject;
	import mx.core.UIComponent;
	import mx.managers.ISystemManager;
	import mx.managers.PopUpManager;
	
	/**
	 * UI用ユーティリティ
	 * @author c9katayama
	 */
	public class UIUtil
	{
		/**
		 * parentの中央にポップアップを追加します。
		 * リサイズ時に中央に移動するイベントリスナーを付与します。
		 * parentがnullの場合、Application.applicationを親とします。
		 * @return popupClassのインスタンス
		 */
		public static function addCenterPopUp(popupWindow:IFlexDisplayObject,modal:Boolean=true,parent:DisplayObject=null):void{
			if(parent==null){
				parent = UIComponent(Application.application);
			}
			lostFocus();
			PopUpManager.addPopUp(popupWindow,parent,modal); 
			addCenterPopUpEventListener(popupWindow,parent);
			PopUpManager.centerPopUp(popupWindow);
		}
		/**
		 * PopUpManager.centerPopUpを呼び出します。
		 * リサイズ時に中央に移動するイベントリスナを付与します。
		 **/
		public static function centerPopUp(popupWindow:IFlexDisplayObject):void{
			addCenterPopUpEventListener(popupWindow,DisplayObject(popupWindow.parent));
			PopUpManager.centerPopUp(popupWindow);
		}
		/**
		 * Popupを作成します。
		 * parentがnullの場合、Application.applicationを親とします。
		 * @return popupClassのインスタンス
		 */
		public static function createPopUp(popupClass:Class,
						modal:Boolean=true,parent:DisplayObject=null):*{
			if(parent==null){
				parent = DisplayObject(Application.application);
			}
			lostFocus();
			return PopUpManager.createPopUp(parent,popupClass,modal);
		}
		private static function lostFocus():void{
			Application(Application.application).setFocus();
		}
		private static function addCenterPopUpEventListener(popup:IFlexDisplayObject,parent:DisplayObject):void{
			var systemManager:ISystemManager;
			if(parent is ISystemManager){
				systemManager = ISystemManager(parent);
			}else{
			 	systemManager = UIComponent(parent).systemManager;
			}			
			var resizeFuntion:Function = function(e:Event):void{
				PopUpManager.centerPopUp(popup);
			};
			var closeFunction:Function = function(e:Event):void{
				systemManager.removeEventListener(Event.RESIZE,resizeFuntion);
				popup.removeEventListener(Event.RESIZE,resizeFuntion);
				popup.removeEventListener(Event.CLOSE,closeFunction);
			};
			systemManager.addEventListener(Event.RESIZE,resizeFuntion);
			popup.addEventListener(Event.RESIZE,resizeFuntion);
			popup.addEventListener(Event.CLOSE,closeFunction);
		}
		/**
		 * ポップアップを、常時最前面に表示されるようにします。
		 */
		public static function alwaysOnTop(popup:IFlexDisplayObject):void{
			if(UIComponent(popup).isPopUp==false){
				return;
			}			
			var f:Function = function(e:Event):void{
				var comp:UIComponent = e.target as UIComponent;
				if(comp != null && comp.isPopUp && (comp is Alert)==false){
					PopUpManager.bringToFront(popup);
				}
			};			
			popup.addEventListener(Event.CLOSE,function(e:*):void{
				Application.application.systemManager.removeEventListener(Event.ADDED,f);
			});
			Application.application.systemManager.addEventListener(Event.ADDED,f);
			PopUpManager.bringToFront(popup);
		}
		
		/**
		 * targetが自分か自分の親オブジェクトかどうかをチェックします。
		 **/	
		public static function isParentOrThis(comp:DisplayObject,target:DisplayObject):Boolean{
			if(comp==target)
				return true;
			return isParent(comp,target);
		}
		/**
		 * targetが自分の親オブジェクトかどうかをチェックします。
		 **/	
		public static function isParent(comp:DisplayObject,target:DisplayObject):Boolean{
			if(!(target is Container)){
				return false;
			}
			var _parent:DisplayObject = comp.parent;
			while(true){
				if(_parent == null){
					break;
				}else{
					if(_parent == target){
						return true;
					}
				}
				_parent = _parent.parent;
			}
			return false;
		}
		/**
		 * 引数のコンポーネントの一番上の親を返します。
		 * 引数がnullの場合、Application.applicationから検索を行います。 
		 **/
		public static function findRoot(component:DisplayObject=null):DisplayObject
		{
			if(component==null){
				component = Application(Application.application);
			}
			var parent:DisplayObject = component;
			while(true){
				if(parent.parent==null){
					break;					
				}else{
					parent = parent.parent;
				}
			}
			return parent;
		}
		/**
		 * 基点となるコンポーネントの親を検索し、直近のPageContainerを検索し、そのIDを返します。
		 */
		public static function findRecentPageContainerId(baseComponent:DisplayObject):String{
			var pageContainerId:String = null;
			while(true){
				if(baseComponent == null){
					throw new Error("PageContainer not found!");
				}
				if(baseComponent is PageContainer){
					pageContainerId = PageContainer(baseComponent).id;
					break;
				}else{
					baseComponent = baseComponent.parent;
				}
			}
			return pageContainerId;			
		}
		/**
		 * 1回のみ実行するイベントリスナーを追加します。 
		 **/
		public static function addOnceEventListener(dispatcher:IEventDispatcher,eventType:String,listener:Function):void{
			EventUtil.addOnceEventListener(dispatcher,eventType,listener);
		}
		/**
		 * 1回のみ実行するイベントリスナーをremoveします。 
		 **/
		public static function removeOnceEventListener(dispatcher:IEventDispatcher,eventType:String,listener:Function):void{
			EventUtil.removeOnceEventListener(dispatcher,eventType,listener);
		}
		/**
		 * 対象とするDisplayObjectがRemoveされた時に、listenerをremoveします。
		 */
		public static function addComponetScopeEventListener(dispatcher:IEventDispatcher,lifeCycleComponent:DisplayObject,type:String,listener:Function):void{
			dispatcher.addEventListener(type,listener);
			UIUtil.callBackWhenComponentRemoved(lifeCycleComponent,function():void{
				ApplicationEventDispatcher.removeEventListener(type,listener);
				dispatcher = null;
				lifeCycleComponent = null;
			});
		}		
		/**
		 * 対象コンポーネントもしくは親コンポーネントがRemoveされた時に、callBackFunctionを呼び出します。
		 */
		public static function callBackWhenComponentRemoved(component:DisplayObject,callBackFunction:Function):void{
			var root:DisplayObject = findRoot(component);
			var parentRemoveFunction:Function = function(e:Event):void{
				if(UIUtil.isParentOrThis(component,e.target as DisplayObject)){
					root.removeEventListener(Event.REMOVED,parentRemoveFunction);
					callBackFunction();
					parentRemoveFunction=null;
					component = null;
					callBackFunction = null;
				}
			};
			root.addEventListener(Event.REMOVED,parentRemoveFunction);
		}
		/**
		 * x,yトを表示オブジェクトの (ローカル) 座標からステージ (グローバル) 座標に変換します。 
		 **/
		public static function localToGlobal(component:DisplayObject,x:Number=0,y:Number=0):Point{
			var p:Point = new Point(x,y);
			p = component.localToGlobal(p);
			return p;
		}
		/**
		 * x,yをステージ (グローバル) 座標から表示オブジェクトの (ローカル) 座標に変換します。
		 */
		public static function globalToLocal(component:DisplayObject,x:Number=0,y:Number=0):Point{
			var p:Point = new Point(x,y);
			p = component.globalToLocal(p);
			return p;
		}
	}
}